
1. [bookmark: _Toc122434485][bookmark: _Toc86739771]Overview
This document lists all the changes needed to correct issues in the ATS iWD-TTCN3-B2020-09_D21wk37 related to the title of this CR.
Contact:	Shaun Harry
	shaun.harry@keysight.com
1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc86739772]Corrections required

[bookmark: _Toc86739773]2.1 Change 1
	Function name
	f_BuildQoSRuleRemoteAccess()

	Reason for change
	QoS rule number 1 is missing (required for test case 7.1.4.1)

	Summary of change
	Add building for QoS rule number 1

	TTCN module
	NG_NAS_Common.ttcn

	MCC160 Comment
	



Before Change:


  function f_BuildQoSRuleRemoteAccess (charstring p_RuleNum) return template (value) QoS_Rule
  {
    var O1_Type v_Id;
    var B1_Type v_DQR;
    var O1_Type v_Precedence;
    var B6_Type v_FlowId;
    var octetstring v_Component;
    var O2_Type v_Length;
    var B4_Type v_FilterId;
    
    select (p_RuleNum) {
      case ("3") {
        v_Id := int2oct(str2int(p_RuleNum),1);
        v_DQR := '0'B;
        v_Precedence := int2oct(str2int(p_RuleNum),1);
        v_FlowId := '000001'B;
        v_FilterId := '0010'B;
        v_Component := fl_BuildRemoteAddressPacketFilter("2");
      }
      case ("4") {
        v_Id := int2oct(str2int(p_RuleNum),1);
        v_DQR := '1'B;
        v_Precedence := int2oct(str2int(p_RuleNum),1);
        v_FlowId := '000010'B;
        v_FilterId := '0011'B;
        v_Component := fl_BuildRemoteAddressPacketFilter("3");
      }
      case ("4a") {
        v_Id := '0F'O;
        v_DQR := '0'B;
        v_Precedence := '0F'O;
        v_FlowId := '000100'B;
        v_FilterId := '1111'B;
        v_Component := fl_BuildRemoteAddressPacketFilter("3a");
      }
      case ("5") {
        v_Id := int2oct(str2int(p_RuleNum),1);
        v_DQR := '0'B;
        v_Precedence := int2oct(str2int(p_RuleNum),1);
        v_FlowId := '000101'B;
        v_FilterId := '0100'B;
        v_Component := fl_BuildRemoteAddressPacketFilter("4");
      }
      case ("6") {
        v_Id := int2oct(str2int(p_RuleNum),1);
        v_DQR := '0'B;
        v_Precedence := int2oct(str2int(p_RuleNum),1);
        v_FlowId := '000110'B;
        v_FilterId := '0101'B;
        v_Component := fl_BuildRemoteAddressPacketFilter("5");
      }
    }
    v_Length := int2oct(lengthof(v_Component) + 5, 2);   // @sic R5s190944 sic@
    return cs_QoS_RuleRemoteAccess(v_Id, v_DQR, v_Precedence, v_FlowId, v_Component, v_Length, v_FilterId);
  }

After Change:

            function f_BuildQoSRuleRemoteAccess (charstring p_RuleNum) return template (value) QoS_Rule
  {
    var O1_Type v_Id;
    var B1_Type v_DQR;
    var O1_Type v_Precedence;
    var B6_Type v_FlowId;
    var octetstring v_Component;
    var O2_Type v_Length;
    var B4_Type v_FilterId;
    
    select (p_RuleNum) {
      case ("1") { 
        v_Id := int2oct(str2int(p_RuleNum),1);
        v_DQR := '1'B;
        v_Precedence  := bit2oct('11111111'B);
        
        v_FlowId := '000001'B;
        v_FilterId := '0000'B;
        v_Component := bit2oct('00000001'B);
      }
       case ("3") {
        v_Id := int2oct(str2int(p_RuleNum),1);
        v_DQR := '0'B;
        v_Precedence := int2oct(str2int(p_RuleNum),1);
        v_FlowId := '000001'B;
        v_FilterId := '0010'B;
        v_Component := fl_BuildRemoteAddressPacketFilter("2");
      }
      case ("4") {
        v_Id := int2oct(str2int(p_RuleNum),1);
        v_DQR := '1'B;
        v_Precedence := int2oct(str2int(p_RuleNum),1);
        v_FlowId := '000010'B;
        v_FilterId := '0011'B;
        v_Component := fl_BuildRemoteAddressPacketFilter("3");
      }
      case ("4a") {
        v_Id := '0F'O;
        v_DQR := '0'B;
        v_Precedence := '0F'O;
        v_FlowId := '000100'B;
        v_FilterId := '1111'B;
        v_Component := fl_BuildRemoteAddressPacketFilter("3a");
      }
      case ("5") {
        v_Id := int2oct(str2int(p_RuleNum),1);
        v_DQR := '0'B;
        v_Precedence := int2oct(str2int(p_RuleNum),1);
        v_FlowId := '000101'B;
        v_FilterId := '0100'B;
        v_Component := fl_BuildRemoteAddressPacketFilter("4");
      }
      case ("6") {
        v_Id := int2oct(str2int(p_RuleNum),1);
        v_DQR := '0'B;
        v_Precedence := int2oct(str2int(p_RuleNum),1);
        v_FlowId := '000110'B;
        v_FilterId := '0101'B;
        v_Component := fl_BuildRemoteAddressPacketFilter("5");
      }
    }
    v_Length := int2oct(lengthof(v_Component) + 5, 2);   // @sic R5s190944 sic@
    return cs_QoS_RuleRemoteAccess(v_Id, v_DQR, v_Precedence, v_FlowId, v_Component, v_Length, v_FilterId);
  }


[bookmark: _Toc86739774]2.2 Change 2
	Function name
	f_TC_7_1_4_1_NR5GC()

	Reason for change
	1. There can only be one default QoS rule defined. Since QoS Rule#4 is set as the Default QoS Rule, QoS Rule #1 cannot also be the default QoS Rule.
2. Following PDU session establishment, ICMPv6 address allocation signalling will occur on QFI 1.

	Summary of change
	1. Changed the DQR bit for the QoSRule#1 to ‘0’ in QoS rules for PDU SESSION ESTABLISHMENT ACCEPT.
Note : An associated prose CR on 38.523-1 will be raised at RAN5#93.
2. Start IP handling on QFI 1.

	TTCN module
	SDAP_NR5GC.ttcn

	MCC160 Comment
	



Before Change:


 function f_TC_7_1_4_1_NR5GC() runs on NR5GC_PTC
  {//SDAP Data Transfer and PDU Header Handling UL/DL
    var template (value) DRB_ToAddModList v_DRB_ToAddModList;
    var GSM_MobilityInfo_Type v_GSM_MobilityInfo_Type;
    var IntegerList_Type v_DRBList;
    var integer v_FirstDRB;
    var integer v_SecondDRB;
    var template (value) QoS_Rules v_QoS_Rules;
    var template (value) QoSFlowDescr v_QoSFlowDescr;
    var template (value) NR_RadioBearerList_Type v_SS_Drb_ConfigList;
    var PDN_Index_Type v_PDN_Index;
    var template (value) QosFlow_Identification_Type v_QosFlowId;
    var integer v_PduSessionId; // @sic R5s191119 sic@
    
    f_NR5GC_Init_NAS(NR_1);

    //Create and configure NR cell1
    f_NR_CellConfig_Def(nr_Cell1);

    //Preamble
    // Perform Registration and get PDU Establishment Request message
    v_GSM_MobilityInfo_Type := f_MultipleQoSPreamble_Part1 (nr_Cell1);
    v_PduSessionId := oct2int(v_GSM_MobilityInfo_Type.SessionId); // @sic R5s191119, R5s200564 sic@
    
    //Get DRB identities
    v_DRBList:= f_NR_GetNextDRBIds(v_GSM_MobilityInfo_Type, true, 2); // First is Default DRB
    v_FirstDRB := v_DRBList[0];
    v_SecondDRB := v_DRBList[1];

    v_SS_Drb_ConfigList := {f_NR_GetRadioBearerConfig_FullAM(nr_Cell1,
                                                             v_SecondDRB,
                                                             -, //RLC normal mode
                                                             -, //PDCP normal mode
                                                             -,
                                                             cs_SDAP_Configuration(v_PduSessionId, Present, Present, {1,2})), //Configure SDAP in normal mode for the moment; TTCN CR 190946
                            f_NR_GetRadioBearerConfig_FullAM(nr_Cell1,
                                                             v_FirstDRB,
                                                             -, //RLC normal mode
                                                             -, //PDCP normal mode
                                                             -,
                                                             cs_SDAP_Configuration(v_PduSessionId, Present, Present, {5,6}))}; //Configure SDAP in normal mode for the moment; TTCN CR 190946
                                                         
    v_DRB_ToAddModList := {cs_38508_NRDRB_ToAddMod(cs_SDAP_Config_Sdap(v_PduSessionId, present_, present_, false, {1,2}), // @sic R5s191119 sic@
                                                   v_SecondDRB,
                                                   cs_38508_PDCP_Config(ms300)),
                           cs_38508_NRDRB_ToAddMod(cs_SDAP_Config_Sdap(v_PduSessionId, present_, present_, true, {5,6}), // @sic R5s191119 sic@
                                                   v_FirstDRB,
                                                   cs_38508_PDCP_Config(ms300))};
                                              
    v_QoS_Rules := cs_QoS_Rules ({f_BuildQoSRuleRemoteAccess("1"), f_BuildQoSRuleRemoteAccess("4"), f_BuildQoSRuleRemoteAccess("5"), f_BuildQoSRuleRemoteAccess("6")}); // @sic R5-215714 sic@
    v_QoSFlowDescr := cs_QoSFlowDescr ({cs_QoS_Flow1, cs_QoS_Flow2, cs_QoS_Flow3, cs_QoS_Flow4});

    // Start IP handling in order for the TTCN to handle IPv6 address allocation,
    // then stop it before the test body, so that TTCN can route the IP data
    v_PDN_Index := f_NR5GC_GetPdnIndex(v_GSM_MobilityInfo_Type);
    v_QosFlowId := cs_IP_QosFlowId(oct2int(v_GSM_MobilityInfo_Type.SessionId), 2); //IP handling will go through QFI 2, because this QFI is associated with the default QoS rule @sic R5s200564, R5s201505 sic@
    f_IP_Handling_StartPDN(IP, v_PDN_Index, cs_DrbInfo_NR(nr_Cell1, v_QosFlowId));
  
    f_MultipleQoSPreamble_Part2 (nr_Cell1,
                                 v_GSM_MobilityInfo_Type,
                                 v_FirstDRB,
                                 v_SecondDRB,
                                 v_DRB_ToAddModList,
                                 v_SS_Drb_ConfigList,
                                 v_QoS_Rules,
                                 v_QoSFlowDescr,
                                 v_PDN_Index); //@sic R5s201361 sic@
                            
    f_NR_AS_CipheringAlgorithm_Set(nea0);

    // Stop IP handling before reconfiguring SDAP to transparent mode
    f_IP_Handling_StopPDN(IP);
    
    // Reconfigure SDAP to transparent mode when moving to connected state
    v_SS_Drb_ConfigList := {f_NR_GetRadioBearerConfig_FullAM(nr_Cell1,
                                                             v_SecondDRB,
                                                             -, //RLC normal mode
                                                             -, //PDCP normal mode
                                                             -,
                                                             cs_SDAP_ConfigTransparentMode(Present)), //Configure SDAP in transparent mode; TTCN CR 190946
                            f_NR_GetRadioBearerConfig_FullAM(nr_Cell1,
                                                             v_FirstDRB,
                                                             -, //RLC normal mode
                                                             -, //PDCP normal mode
                                                             -,
                                                             cs_SDAP_ConfigTransparentMode(Present))}; //Configure SDAP in transparent mode; TTCN CR 190946
    
    f_NR5GC_RRC_ConnectedState3N_Def(nr_Cell1,
                                     TEST_LOOPModeB_ON,
                                     -,
                                     '01'O, //IP PDU delay  = 1 second
                                     -,-,
                                     v_DRB_ToAddModList,
                                     v_SS_Drb_ConfigList);
    
    f_NR_TestBody_Set(true);
    fl_TC_7_1_4_1_NR_TestBody(v_FirstDRB, v_SecondDRB, v_GSM_MobilityInfo_Type);
    f_NR_TestBody_Set(false);
    
    f_NR5GC_OpenUE_TestLoopMode_Deactivate_TestMode(nr_Cell1); // @sic R5s191119 sic@
    
    //Postamble
    f_NR_Postamble(nr_Cell1, STATE_CONNECTED_3A);
    f_NR_ReleaseAllCells();
  }
  
After Change:

function f_TC_7_1_4_1_NR5GC() runs on NR5GC_PTC
  {//SDAP Data Transfer and PDU Header Handling UL/DL
    var template (value) DRB_ToAddModList v_DRB_ToAddModList;
    var GSM_MobilityInfo_Type v_GSM_MobilityInfo_Type;
    var IntegerList_Type v_DRBList;
    var integer v_FirstDRB;
    var integer v_SecondDRB;
    var template (value) QoS_Rules v_QoS_Rules;
    var template (value) QoSFlowDescr v_QoSFlowDescr;
    var template (value) NR_RadioBearerList_Type v_SS_Drb_ConfigList;
    var PDN_Index_Type v_PDN_Index;
    var template (value) QosFlow_Identification_Type v_QosFlowId;
    var integer v_PduSessionId; // @sic R5s191119 sic@
    
    f_NR5GC_Init_NAS(NR_1);

    //Create and configure NR cell1
    f_NR_CellConfig_Def(nr_Cell1);

    //Preamble
    // Perform Registration and get PDU Establishment Request message
    v_GSM_MobilityInfo_Type := f_MultipleQoSPreamble_Part1 (nr_Cell1);
    v_PduSessionId := oct2int(v_GSM_MobilityInfo_Type.SessionId); // @sic R5s191119, R5s200564 sic@
    
    //Get DRB identities
    v_DRBList:= f_NR_GetNextDRBIds(v_GSM_MobilityInfo_Type, true, 2); // First is Default DRB
    v_FirstDRB := v_DRBList[0];
    v_SecondDRB := v_DRBList[1];

    v_SS_Drb_ConfigList := {f_NR_GetRadioBearerConfig_FullAM(nr_Cell1,
                                                             v_SecondDRB,
                                                             -, //RLC normal mode
                                                             -, //PDCP normal mode
                                                             -,
                                                             cs_SDAP_Configuration(v_PduSessionId, Present, Present, {1,2})), //Configure SDAP in normal mode for the moment; TTCN CR 190946
                            f_NR_GetRadioBearerConfig_FullAM(nr_Cell1,
                                                             v_FirstDRB,
                                                             -, //RLC normal mode
                                                             -, //PDCP normal mode
                                                             -,
                                                             cs_SDAP_Configuration(v_PduSessionId, Present, Present, {5,6}))}; //Configure SDAP in normal mode for the moment; TTCN CR 190946
                                                         
    v_DRB_ToAddModList := {cs_38508_NRDRB_ToAddMod(cs_SDAP_Config_Sdap(v_PduSessionId, present_, present_, false, {1,2}), // @sic R5s191119 sic@
                                                   v_SecondDRB,
                                                   cs_38508_PDCP_Config(ms300)),
                           cs_38508_NRDRB_ToAddMod(cs_SDAP_Config_Sdap(v_PduSessionId, present_, present_, true, {5,6}), // @sic R5s191119 sic@
                                                   v_FirstDRB,
                                                   cs_38508_PDCP_Config(ms300))};
                                              
    v_QoS_Rules := cs_QoS_Rules ({f_BuildQoSRuleRemoteAccess("1"), f_BuildQoSRuleRemoteAccess("4"), f_BuildQoSRuleRemoteAccess("5"), f_BuildQoSRuleRemoteAccess("6")}); // @sic R5-215714 sic@
 
    v_QoS_Rules[0].dqrBit := '0'B;

    v_QoSFlowDescr := cs_QoSFlowDescr ({cs_QoS_Flow1, cs_QoS_Flow2, cs_QoS_Flow3, cs_QoS_Flow4});

    // Start IP handling in order for the TTCN to handle IPv6 address allocation,
    // then stop it before the test body, so that TTCN can route the IP data
    v_PDN_Index := f_NR5GC_GetPdnIndex(v_GSM_MobilityInfo_Type);
    v_QosFlowId := cs_IP_QosFlowId(oct2int(v_GSM_MobilityInfo_Type.SessionId), 1); //IP handling will go through QFI 2, because this QFI is associated with the default QoS rule @sic R5s200564, R5s201505 sic@
    f_IP_Handling_StartPDN(IP, v_PDN_Index, cs_DrbInfo_NR(nr_Cell1, v_QosFlowId));
  
    f_MultipleQoSPreamble_Part2 (nr_Cell1,
                                 v_GSM_MobilityInfo_Type,
                                 v_FirstDRB,
                                 v_SecondDRB,
                                 v_DRB_ToAddModList,
                                 v_SS_Drb_ConfigList,
                                 v_QoS_Rules,
                                 v_QoSFlowDescr,
                                 v_PDN_Index); //@sic R5s201361 sic@
                            
    f_NR_AS_CipheringAlgorithm_Set(nea0);

    // Stop IP handling before reconfiguring SDAP to transparent mode
    f_IP_Handling_StopPDN(IP);
    
    // Reconfigure SDAP to transparent mode when moving to connected state
    v_SS_Drb_ConfigList := {f_NR_GetRadioBearerConfig_FullAM(nr_Cell1,
                                                             v_SecondDRB,
                                                             -, //RLC normal mode
                                                             -, //PDCP normal mode
                                                             -,
                                                             cs_SDAP_ConfigTransparentMode(Present)), //Configure SDAP in transparent mode; TTCN CR 190946
                            f_NR_GetRadioBearerConfig_FullAM(nr_Cell1,
                                                             v_FirstDRB,
                                                             -, //RLC normal mode
                                                             -, //PDCP normal mode
                                                             -,
                                                             cs_SDAP_ConfigTransparentMode(Present))}; //Configure SDAP in transparent mode; TTCN CR 190946
    
    f_NR5GC_RRC_ConnectedState3N_Def(nr_Cell1,
                                     TEST_LOOPModeB_ON,
                                     -,
                                     '01'O, //IP PDU delay  = 1 second
                                     -,-,
                                     v_DRB_ToAddModList,
                                     v_SS_Drb_ConfigList);
    
    f_NR_TestBody_Set(true);
    fl_TC_7_1_4_1_NR_TestBody(v_FirstDRB, v_SecondDRB, v_GSM_MobilityInfo_Type);
    f_NR_TestBody_Set(false);
    
    f_NR5GC_OpenUE_TestLoopMode_Deactivate_TestMode(nr_Cell1); // @sic R5s191119 sic@
    
    //Postamble
    f_NR_Postamble(nr_Cell1, STATE_CONNECTED_3A);
    f_NR_ReleaseAllCells();
  }
  

[bookmark: _Toc86739775]2.3 Change 3
	Function name
	fl_TC_7_1_4_1_NR_TestBody()

	Reason for change
	At steps 1 and 2 IP datagram should be created using filter#3 to use QFI 2 

	Summary of change
	Create datagram for QFI 2, for steps 1 and 2.

	TTCN module
	SDAP_NR5GC.ttcn

	MCC160 Comment
	



Before Change:


 function fl_TC_7_1_4_1_NR_TestBody(integer p_FirstDRB,
                                     integer p_SecondDRB,
                                     GSM_MobilityInfo_Type p_GSM_MobilityInfo_Type) runs on NR_BASE_PTC
  {
    var template (present) SDAP_PDU_Type v_ExpectedPdu;
    var octetstring v_IPDataQFI1 := f_SdapTests_CreateDatagram("2"); // Packet with remote address matching filter #2
    var octetstring v_IPDataQFI5 := f_SdapTests_CreateDatagram("4"); // Packet with remote address matching filter #4
    var octetstring v_IPDataQFI4 := f_SdapTests_CreateDatagram("3a"); // Packet with remote address matching filter #3a
    var template (value) NG_NAS_DL_Pdu_Type v_PiggyBackedMsgToSend;
    var template (value) DRB_ToAddModList v_DRB_ToAddModList;
    var template (value) RadioBearerConfig v_RadioBearerConfig;
    var integer v_PduSessionId := oct2int(p_GSM_MobilityInfo_Type.SessionId);  // @sic R5s191119, R5s200564 sic@
    var boolean v_PDU := false; // @sic R5s191119 sic@; boolean to indicate if the SDAP PDU has been received
    var boolean v_EndMarker := false; // @sic R5s191119 sic@; boolean to indicate if the End-Marker has been received
    var boolean v_RrcMsg := false; // @sic R5s191119 sic@; boolean to indicate if the RRC message has been received
    var boolean v_NasMsg := false; // @sic R5s191119 sic@; boolean to indicate if the NAS message has been received
    var NR_SRB_COMMON_IND v_ReceivedAsp; // @sic R5s191119 sic@
    
    // @siclog "Step 1" siclog@
    // The SS sends the SDAP Data PDU with SDAP header on DRB 2 and the following content to the UE: RDI=0. RQI=0, QFI=2.
    f_SDAP_SendWithHeader(nr_Cell1, p_SecondDRB, tsc_RDI0, tsc_RQI0, tsc_QFI2, v_IPDataQFI1); //@sic R5-215714 sic@
        
    // @siclog "Step 2" siclog@
    // Check: Does the UE re-transmit SDAP Data PDU on DRB 2 with SDAP header as per the stored DRB mapping flow with QFI=1?
    v_ExpectedPdu := cr_SDAP_PDU_UL_WithHeader(cr_SDAP_UL_PduHeader_Data(tsc_QFI2),v_IPDataQFI1); //@sic R5-215714 sic@
    DRB.receive(car_NR_DRB_COMMON_IND_SDAP_PDUList(nr_Cell1, p_SecondDRB, -, {v_ExpectedPdu}));
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 2");

    // @siclog "Step 3" siclog@
    // The SS sends the SDAP Data PDU with SDAP header on DRB 2 and the following content to the UE: RDI=1, RQI=0, QFI=5.
    f_SDAP_SendWithHeader(nr_Cell1, p_SecondDRB, tsc_RDI1, tsc_RQI0, tsc_QFI5, v_IPDataQFI5);
    
    // EXCEPTION: In parallel to the event described in step 4 the events specified in Table 7.1.4.1.3.2-2 shall take place.
    // @siclog "Table 7.1.4.1.3.2-2, Step 1" siclog@
    // @siclog "Step 4" siclog@

    v_ExpectedPdu := cr_SDAP_PDU_UL_WithHeader(cr_SDAP_UL_PduHeader_Data(tsc_QFI5),v_IPDataQFI5);
    // @sic R5s191119 sic@
    do {
      alt {
        [not v_PDU] DRB.receive(car_NR_DRB_COMMON_IND_SDAP_PDUList(nr_Cell1, p_SecondDRB, -, {v_ExpectedPdu}))
          { // Check: Does the UE re-transmit SDAP Data PDU on DRB 2 with SDAP header as per the stored DRB mapping Flow with QFI=5?
            v_PDU := true;
            f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 4");
          }
        [not v_EndMarker] DRB.receive(car_NR_DRB_COMMON_IND_SDAP_PDUList(nr_Cell1, p_FirstDRB, -, {cr_SDAP_PDU_UL_EndMarker(tsc_QFI5)}))
          { // Check: Does the UE transmits End-Marker Control PDU on DRB1 for QFI=5?
            v_EndMarker := true;
            f_NR_PreliminaryPass(__FILE__, __LINE__, "End-Marker received");
          }
      }
    }
    while (not(v_PDU and v_EndMarker));
    
    // @siclog "Step 5" siclog@
    // The SS transmits an RRCReconfiguration message including a PDU SESSION MODIFICATION COMMAND
    // Note: As SDAP is in transparent mode, there is no need to reconfigure the SS
    v_DRB_ToAddModList := {cs_38508_NRDRB_ToAddMod(cs_SDAP_Config_Sdap(v_PduSessionId, present_, present_, false, {4}), // @sic R5s191119 sic@
                                                   p_SecondDRB,
                                                   cs_38508_PDCP_Config(ms300))};
   
    v_RadioBearerConfig := cs_38508_RadioBearerConfigDef(omit,
                                                         v_DRB_ToAddModList,
                                                         omit,
                                                         cs_38508_SecurityConfig(cs_38508_SecurityAlgorithmConfig(f_NR_AS_CipheringAlgorithm_Get(), f_NR_AS_IntegrityAlgorithm_Get()), master));
                                                         
    v_PiggyBackedMsgToSend := f_Get_NG_PDUSessionModificationCommand (p_GSM_MobilityInfo_Type.SessionId,
                                                                      -,-,-,-,-,
                                                                      cs_QoS_Rules ({f_BuildQoSRuleRemoteAccess("4a")}, '7A'O), // @sic R5s191119 sic@
                                                                      cs_QoSFlowDescr ({cs_QoS_Flow2a}));
   
    SRB.send(cas_NR_RrcNasPduList_REQ(nr_Cell1,
                                      tsc_NR_RbId_SRB1,
                                      cs_TimingInfo_Now,
                                      cs_NR_RRC_MSG_RequestDCCH(cs_38508_RRCReconfiguration(tsc_NR_RRC_TI_Def, v_RadioBearerConfig)),
                                      {cs_NG_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered, v_PiggyBackedMsgToSend)}));
    
    // EXCEPTION: In parallel to the event described in step 6 the events specified in Tables 7.1.4.1.3.2-3 and 7.1.4.1.3.2-4 shall take place.
    // @siclog "Table 7.1.4.1.3.3-3, Step 1" siclog@
    // @siclog "Table 7.1.4.1.3.3-4, Step 1" siclog@
    // @siclog "Step 6" siclog@
    // @sic R5s191119 sic@
    v_EndMarker := false; // Reset variable for reusing
    do {
      alt {
        [not v_RrcMsg] SRB.receive(car_NR_SRB1_RrcPdu_IND(nr_Cell1, cr_38508_RRCReconfigurationComplete))
          { // The UE transmits an RRCReconfigurationComplete message.
            v_RrcMsg := true;
            f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 6");
          }
        [not v_EndMarker] DRB.receive(car_NR_DRB_COMMON_IND_SDAP_PDUList(nr_Cell1, p_FirstDRB, -, {cr_SDAP_PDU_UL_EndMarker(tsc_QFI4)}))
          { // Check: Does the UE transmit End-Marker Control PDU on DRB 1 for QFI=4?
            v_EndMarker := true;
            f_NR_PreliminaryPass(__FILE__, __LINE__, "End-Marker received");
          }
        [not v_NasMsg] SRB.receive(car_NR_SRB_NasPdu_IND(nr_Cell1, tsc_NR_RbId_SRB2, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp
          { // The UE Transmits PDU SESSION MODIFICATION COMPLETE
            if (match (v_ReceivedAsp.Signalling.Nas[0].Pdu.PiggybackedPduList[0].Msg , cr_NG_PDU_SESSION_MODIFICATION_COMPLETE(?,?))) {
              v_NasMsg := true;
            }
            else {
              f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Unexpected NAS message received");
            }
          }
      }
    }
    while (not(v_RrcMsg and v_EndMarker and v_NasMsg));
    
    // @siclog "Step 7" siclog@
    // The SS sends the SDAP Data PDU with SDAP header on DRB 2 and the following content to the UE: RDI=0. RQI=0, QFI=4.
    f_SDAP_SendWithHeader(nr_Cell1, p_SecondDRB, tsc_RDI0, tsc_RQI0, tsc_QFI4, v_IPDataQFI4);
    
    // @siclog "Step 8" siclog@
    // Check: Does the UE re-transmit SDAP Data PDU on DRB 2 with SDAP header as per the stored DRB mapping Flow with QFI=4?
    v_ExpectedPdu := cr_SDAP_PDU_UL_WithHeader(cr_SDAP_UL_PduHeader_Data(tsc_QFI4),v_IPDataQFI4);
    DRB.receive(car_NR_DRB_COMMON_IND_SDAP_PDUList(nr_Cell1, p_SecondDRB, -, {v_ExpectedPdu}));
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 8");
      
  }
  
After Change:

 function fl_TC_7_1_4_1_NR_TestBody(integer p_FirstDRB,
                                     integer p_SecondDRB,
                                     GSM_MobilityInfo_Type p_GSM_MobilityInfo_Type) runs on NR_BASE_PTC
  {
    var template (present) SDAP_PDU_Type v_ExpectedPdu;
    var octetstring v_IPDataQFI1 := f_SdapTests_CreateDatagram("2"); // Packet with remote address matching filter #2
    var octetstring v_IPDataQFI2 := f_SdapTests_CreateDatagram("3"); // Packet with remote address matching filter #3
    var octetstring v_IPDataQFI5 := f_SdapTests_CreateDatagram("4"); // Packet with remote address matching filter #4
    var octetstring v_IPDataQFI4 := f_SdapTests_CreateDatagram("3a"); // Packet with remote address matching filter #3a
    var template (value) NG_NAS_DL_Pdu_Type v_PiggyBackedMsgToSend;
    var template (value) DRB_ToAddModList v_DRB_ToAddModList;
    var template (value) RadioBearerConfig v_RadioBearerConfig;
    var integer v_PduSessionId := oct2int(p_GSM_MobilityInfo_Type.SessionId);  // @sic R5s191119, R5s200564 sic@
    var boolean v_PDU := false; // @sic R5s191119 sic@; boolean to indicate if the SDAP PDU has been received
    var boolean v_EndMarker := false; // @sic R5s191119 sic@; boolean to indicate if the End-Marker has been received
    var boolean v_RrcMsg := false; // @sic R5s191119 sic@; boolean to indicate if the RRC message has been received
    var boolean v_NasMsg := false; // @sic R5s191119 sic@; boolean to indicate if the NAS message has been received
    var NR_SRB_COMMON_IND v_ReceivedAsp; // @sic R5s191119 sic@
    
    // @siclog "Step 1" siclog@
    // The SS sends the SDAP Data PDU with SDAP header on DRB 2 and the following content to the UE: RDI=0. RQI=0, QFI=2.
    f_SDAP_SendWithHeader(nr_Cell1, p_SecondDRB, tsc_RDI0, tsc_RQI0, tsc_QFI2, v_IPDataQFI2); //@sic R5-215714 sic@
        
    // @siclog "Step 2" siclog@
    // Check: Does the UE re-transmit SDAP Data PDU on DRB 2 with SDAP header as per the stored DRB mapping flow with QFI=1?
    v_ExpectedPdu := cr_SDAP_PDU_UL_WithHeader(cr_SDAP_UL_PduHeader_Data(tsc_QFI2),v_IPDataQFI2); //@sic R5-215714 sic@
    DRB.receive(car_NR_DRB_COMMON_IND_SDAP_PDUList(nr_Cell1, p_SecondDRB, -, {v_ExpectedPdu}));
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 2");

    // @siclog "Step 3" siclog@
    // The SS sends the SDAP Data PDU with SDAP header on DRB 2 and the following content to the UE: RDI=1, RQI=0, QFI=5.
    f_SDAP_SendWithHeader(nr_Cell1, p_SecondDRB, tsc_RDI1, tsc_RQI0, tsc_QFI5, v_IPDataQFI5);
    
    // EXCEPTION: In parallel to the event described in step 4 the events specified in Table 7.1.4.1.3.2-2 shall take place.
    // @siclog "Table 7.1.4.1.3.2-2, Step 1" siclog@
    // @siclog "Step 4" siclog@

    v_ExpectedPdu := cr_SDAP_PDU_UL_WithHeader(cr_SDAP_UL_PduHeader_Data(tsc_QFI5),v_IPDataQFI5);
    // @sic R5s191119 sic@
    do {
      alt {
        [not v_PDU] DRB.receive(car_NR_DRB_COMMON_IND_SDAP_PDUList(nr_Cell1, p_SecondDRB, -, {v_ExpectedPdu}))
          { // Check: Does the UE re-transmit SDAP Data PDU on DRB 2 with SDAP header as per the stored DRB mapping Flow with QFI=5?
            v_PDU := true;
            f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 4");
          }
        [not v_EndMarker] DRB.receive(car_NR_DRB_COMMON_IND_SDAP_PDUList(nr_Cell1, p_FirstDRB, -, {cr_SDAP_PDU_UL_EndMarker(tsc_QFI5)}))
          { // Check: Does the UE transmits End-Marker Control PDU on DRB1 for QFI=5?
            v_EndMarker := true;
            f_NR_PreliminaryPass(__FILE__, __LINE__, "End-Marker received");
          }
      }
    }
    while (not(v_PDU and v_EndMarker));
    
    // @siclog "Step 5" siclog@
    // The SS transmits an RRCReconfiguration message including a PDU SESSION MODIFICATION COMMAND
    // Note: As SDAP is in transparent mode, there is no need to reconfigure the SS
    v_DRB_ToAddModList := {cs_38508_NRDRB_ToAddMod(cs_SDAP_Config_Sdap(v_PduSessionId, present_, present_, false, {4}), // @sic R5s191119 sic@
                                                   p_SecondDRB,
                                                   cs_38508_PDCP_Config(ms300))};
   
    v_RadioBearerConfig := cs_38508_RadioBearerConfigDef(omit,
                                                         v_DRB_ToAddModList,
                                                         omit,
                                                         cs_38508_SecurityConfig(cs_38508_SecurityAlgorithmConfig(f_NR_AS_CipheringAlgorithm_Get(), f_NR_AS_IntegrityAlgorithm_Get()), master));
                                                         
    v_PiggyBackedMsgToSend := f_Get_NG_PDUSessionModificationCommand (p_GSM_MobilityInfo_Type.SessionId,
                                                                      -,-,-,-,-,
                                                                      cs_QoS_Rules ({f_BuildQoSRuleRemoteAccess("4a")}, '7A'O), // @sic R5s191119 sic@
                                                                      cs_QoSFlowDescr ({cs_QoS_Flow2a}));
   
    SRB.send(cas_NR_RrcNasPduList_REQ(nr_Cell1,
                                      tsc_NR_RbId_SRB1,
                                      cs_TimingInfo_Now,
                                      cs_NR_RRC_MSG_RequestDCCH(cs_38508_RRCReconfiguration(tsc_NR_RRC_TI_Def, v_RadioBearerConfig)),
                                      {cs_NG_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered, v_PiggyBackedMsgToSend)}));
    
    // EXCEPTION: In parallel to the event described in step 6 the events specified in Tables 7.1.4.1.3.2-3 and 7.1.4.1.3.2-4 shall take place.
    // @siclog "Table 7.1.4.1.3.3-3, Step 1" siclog@
    // @siclog "Table 7.1.4.1.3.3-4, Step 1" siclog@
    // @siclog "Step 6" siclog@
    // @sic R5s191119 sic@
    v_EndMarker := false; // Reset variable for reusing
    do {
      alt {
        [not v_RrcMsg] SRB.receive(car_NR_SRB1_RrcPdu_IND(nr_Cell1, cr_38508_RRCReconfigurationComplete))
          { // The UE transmits an RRCReconfigurationComplete message.
            v_RrcMsg := true;
            f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 6");
          }
        [not v_EndMarker] DRB.receive(car_NR_DRB_COMMON_IND_SDAP_PDUList(nr_Cell1, p_FirstDRB, -, {cr_SDAP_PDU_UL_EndMarker(tsc_QFI4)}))
          { // Check: Does the UE transmit End-Marker Control PDU on DRB 1 for QFI=4?
            v_EndMarker := true;
            f_NR_PreliminaryPass(__FILE__, __LINE__, "End-Marker received");
          }
        [not v_NasMsg] SRB.receive(car_NR_SRB_NasPdu_IND(nr_Cell1, tsc_NR_RbId_SRB2, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp
          { // The UE Transmits PDU SESSION MODIFICATION COMPLETE
            if (match (v_ReceivedAsp.Signalling.Nas[0].Pdu.PiggybackedPduList[0].Msg , cr_NG_PDU_SESSION_MODIFICATION_COMPLETE(?,?))) {
              v_NasMsg := true;
            }
            else {
              f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Unexpected NAS message received");
            }
          }
      }
    }
    while (not(v_RrcMsg and v_EndMarker and v_NasMsg));
    
    // @siclog "Step 7" siclog@
    // The SS sends the SDAP Data PDU with SDAP header on DRB 2 and the following content to the UE: RDI=0. RQI=0, QFI=4.
    f_SDAP_SendWithHeader(nr_Cell1, p_SecondDRB, tsc_RDI0, tsc_RQI0, tsc_QFI4, v_IPDataQFI4);
    
    // @siclog "Step 8" siclog@
    // Check: Does the UE re-transmit SDAP Data PDU on DRB 2 with SDAP header as per the stored DRB mapping Flow with QFI=4?
    v_ExpectedPdu := cr_SDAP_PDU_UL_WithHeader(cr_SDAP_UL_PduHeader_Data(tsc_QFI4),v_IPDataQFI4);
    DRB.receive(car_NR_DRB_COMMON_IND_SDAP_PDUList(nr_Cell1, p_SecondDRB, -, {v_ExpectedPdu}));
    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 8");
      
  }



